

# Grid Square Location Aid

*This program aims to please.*

by C. L. Houghton WB6IGP

Operation on any of our microwave bands above 2 GHz requires a means to determine compass headings to and from the station with which you want to make contact. Since most antennas in use at these frequencies are very highly directive, you need to know precisely where to point them to make contact with the distant station.

On short contacts of 10 to 15 miles, the pointing angle is not too critical. An error of 10 to 20 degrees will still provide reasonable signal strength when in the clear. When the path to cover is not line-of-sight, however, you need some pointing aid.

A common method is to lay out the pointing angle on a map with a ruler and determine the

---

***"The program  
will run on virtually  
any BASIC operation  
system."***

---

prospective compass heading to and from each location. This works out well, but requires quite a bit of time to figure out all needed compass headings. For contest time, a faster method is needed to determine distance points between contacts.

## Location Finder

The solution to this problem is a computer program (see list in sidebar) that provides you with the compass headings from and to the desired location of contact. In addition, it determines the distance between locations in miles and kilometers. This program was adapted from multiple sources by Leon Helms WA5BNH. The program will default to your location by changing the statement in sequence 30 and 40 to your latitude (30) and longitude (40). The program can bypass the default location and you can input specific information for lati-

```

10 REM ASSEMBLED FROM MULTIPLE SOURCES BY LEON HELMS WA5BNH 9/15/86
20 PRINT:PRINT
25 MY$="MT OTAY"
30 DA=32.5958334#
40 DG=-116.8458334#
50 GOSUB 1000
60 PRINT"Do you wish to use lat/long or grid squares? (L/G)";
70 INPUT A$:IF A$ = "l" OR A$ = "L" THEN L=1:GOTO 100
80 IF A$ = "g" OR A$ = "G" THEN L=0:GOTO 100
90 GOTO 60
100 PRINT "Do you wish to use the default coordinates for your location"
101 IF L = 1 GOTO 110
102 LA=DA:LO=DG:GOSUB 1500
105 PRINT " ";MY$;" ";G$;"
107 GOTO 130
110 XX=DA:GOSUB 1600
114 PRINT XD:CHR$(248);XM:"";XN:CHR$(34);" ";
116 PRINT N$:" X ";:XX=DG:GOSUB 1600
118 PRINT XD:CHR$(248);XM:"";XN:CHR$(34);" ";
122 PRINT W$:" (y/n)";
130 INPUT A$:IF A$ = "y" OR A$ = "Y" THEN GOTO 190
135 IF A$="" THEN GOTO 190
140 IF A$ <> "n" AND A$ <> "N" THEN GOTO 100
150 PRINT:PRINT"Enter your location"
160 GOSUB 1100
170 DA = A2:DG = L2
180 GOSUB 1000
190 PRINT:PRINT"Enter other location name"
195 INPUT NA$
200 GOSUB 1100
210 K1 = L2:K2 = A2
220 PI = 3.1415926#
230 A1 = DA*PI/180
240 A2 = A2*PI/180
250 N = 1
260 L1 = DG*PI/180
270 L2 = L2*PI/180
280 Q2=TAN(A1)*COS(L2-L1)-TAN(A2)
290 Q1=COS(A1)*(Q2/SIN(L2-L1))
300 X1=270+((180/PI)*ATN(Q1))
310 L9=L2-L1
320 IF L9 <=(-PI) THEN 360
330 IF L9 <=0 THEN 370
340 IF L9 <=PI THEN 360
350 GOTO 370
360 X1=X1-180
370 IF U = 2 THEN 490
380 L3=L2
390 L4=L1
400 L2=L4
410 L1=L3
420 A3=A2
430 A4=A1
440 A2=A4
450 A1=A3
460 Y1=X1
470 N=2
480 GOTO 280
490 C1=ABS(L9)
500 IF C1>PI THEN 520
510 GOTO 540
520 C1=2*PI-C1
530 REM arc cosine must be converted to equivalent form in basic
540 Z2=SIN(A1)*SIN(A2)
550 Z3=COS(A1)*COS(A2)*COS(C1)
560 Z4=Z2+Z3
570 Z1=-ATN(Z4/SQR(-Z4*Z4+1))+(PI/2)
580 Z1=Z1*180/PI
590 A=69.05*Z1
600 B=60*Z1

```

*The BASIC listing of the grid square location program.*

```

610 E=111.12*Z1
630 C=Y1+180
640 IF C>360 THEN C=Y1-180
650 Y1=INT(Y1*100+.5)/100
660 X1=INT(X1*100+.5)/100
670 A=INT(A*100+.5)/100
680 B=INT(B*100+.5)/100
690 IF K2 > 0 THEN S$ = "N" ELSE S$ = "S"
700 IF K1 > 0 THEN E$ = "E" ELSE E$ = "W"
705 GOSUB 1000
710 XX=DA:GOSUB 1600
720 PRINT XD;CHR$(248);XM;"";XN;CHR$(34);" ";
730 PRINT N$;" ";XX=DG:GOSUB 1600
740 PRINT XD;CHR$(248);XM;"";XN;CHR$(34);" ";
750 PRINT W$;
760 PRINT TAB(39);"to";
770 XX=K2:GOSUB 1600
780 PRINT TAB(43);
790 PRINT XD;CHR$(248);XM;"";XN;CHR$(34);" ";
800 PRINT S$;" ";
810 XX=K1:GOSUB 1600
820 PRINT XD;CHR$(248);XM;"";XN;CHR$(34);" ";
830 PRINT W$
840 LA=DA;LO=DG:GOSUB 1500
850 PRINT " ";MY$;" ";G$;TAB(39);"to ";NA$;TAB(60);
860 LA=K2;LO=K1:GOSUB 1500
870 PRINT G$
880 PRINT
890 PRINT"Local true heading =\"Y1;\"deg\";TAB(40);
900 E=INT(E*100+.5)/100
910 PRINT"Distant true heading =\"X1;\"deg\"
920 PRINT"Statute miles =\";A
930 PRINT"Nautical miles =\";B
940 PRINT"Kilometers =\";E
960 GOTO 190
1000 IF DA > 0 THEN N$ = "N" ELSE N$ = "S"
1010 IF DG > 0 THEN W$ = "E" ELSE W$ = "W"
1020 RETURN
1100 IF L = 0 THEN GOTO 1300 ELSE GOTO 1120
1110 PRINT"Please specify 'N' or 'S'. "
1120 PRINT"Enter latitude: Degrees,minutes,seconds,(N or S)"
1130 INPUT W1,W2,W3,I$
1140 IF I$ <> "n" AND I$ <> "s" AND I$ <> "N" AND I$ <> "S" THEN GOTO 1110
1150 IF I$ = "s" OR I$ = "S" THEN W1 = -W1;W2 = -W2;W3 = -W3
1160 A2 = W1 + (W2/60) + (W3/3600)
1170 GOTO 1190
1180 PRINT"Please specify 'E' or 'W'. "
1190 PRINT"Enter longitude: Degrees,minutes,seconds,(E or W)"
1200 INPUT R1,R2,R3,D$
1210 IF D$ <> "e" AND D$ <> "w" AND D$ <> "E" AND D$ <> "W" THEN GOTO 1180
1220 IF D$ = "w" OR D$ = "W" THEN R1 = -R1;R2 = -R2;R3 = -R3
1230 L2 = R1 + (R2/60) + (R3/3600)
1240 RETURN
1300 INPUT "Grid square (ex. DM13ET) ";C$
1310 IF LEN(C$)=4 THEN C$=C$+"MM"
1320 ER=0:IF LEN(C$)<6 THEN GOTO 1300
1330 FOR K=1 TO 6:A(K)=ASC(MID$(C$,K,1)):NEXT
1340 IF A(1)>90 THEN A(1)=A(1)-32
1350 IF A(2)>90 THEN A(2)=A(2)-32
1360 IF A(5)>90 THEN A(5)=A(5)-32
1370 IF A(6)>90 THEN A(6)=A(6)-32
1380 L2=-((180-(A(1)-65)*20-(A(3)-48)*2-(A(5)-64.5)/12)
1390 A2=-90+(A(2)-65)*10+(A(4)-48)+(A(6)-64.6)/24
1400 RETURN
1500 IF INT(LA/10)=LA/10 THEN GOTO 1700
1503 IF INT(LO/20)=LO/20 THEN GOTO 1700
1505 B(2)=INT((LA+90)/10)
1510 B(4)=INT((LA+90)-(B(2)*10))
1520 B(6)=INT((LA-INT(LA))*24)+65
1530 B(1)=INT((180+LO)/20)
1540 B(3)=INT(((LO+180)-(B(1)*20))/2)
1550 B(5)=INT((LO-INT(LO))*12)+65
1560 IF INT(ABS(LO)/2) = INT(ABS(LO))/2 THEN B(5)=B(5)+12
1570 G$=CHR$(B(1)+65)+CHR$(B(2)+65)+CHR$(B(3)+48)+CHR$(B(4)+48)+CHR$(B(5))
1580 G$=G$+CHR$(B(6))
1590 RETURN
1600 XA=ABS(XX)
1610 XD=INT(XA)
1620 XM=INT((XA-XD)*60)
1630 XN=INT((XA-(XD+(XM/60)))*3600)
1640 RETURN
1700 N=0:L=0:O=0
1705 IF INT(LA/10) = LA/10 THEN N = 1:L = 1
1710 IF INT(LO/20) = LO/20 THEN O = 1:L = L+1
1720 IF N=1 THEN LA=LA+1/3600 :REM add one second
1730 IF O=1 THEN LO=LO+1/3600 :REM add one second
1740 GOSUB 1505
1750 CC=POS(C) :REM get the cursor position
1760 PRINT G$ :REM print on possible answer
1770 PRINT TAB(CC); :REM TAB BACK TO WHERE THE CURSOR WAS
1775 IF N=1 THEN LA=LA-2/3600
1780 IF L = 1 GOTO 1820
1790 GOSUB 1505
1800 PRINT G$ :REM print on possible answer
1810 PRINT TAB(CC); :REM TAB BACK TO WHERE THE CURSOR WAS
1820 IF O=1 THEN LO=LO-2/3600
1825 IF L = 1 GOTO 1505
1830 GOSUB 1505
1840 CC=POS(C) :REM get the cursor position
1850 PRINT G$ :REM print on possible answer
1860 PRINT TAB(CC); :REM TAB BACK TO WHERE THE CURSOR WAS
1870 LA=LA+2/3600
1880 GOTO 1505

```

The BASIC listing of the grid square location program (continued).

tude/longitude or grid squares, as it is all menu driven.

I used the program to assist a contact during the last 10 GHz microwave contest from Mt. Helix, in San Diego proper, to Heaps Peak, located near Lake Arrowhead in the mountains above San Bernardino. Entering the program, I selected the grid square option and omitted the default location and replaced it with my location at Mt. Helix—grid square DM12LT. The other station, contacted on the co-ordinating frequency on 2, stated he was located at DM14KF, Heaps Peak. The computer (Tandy 100) took 5 seconds to produce the following output:  
DM12LT 32 X 48.30 N, 117 X 2.29 W TO  
DM14KF 34 X 13.29 S, 117 X 7.29 W  
LOCAL TRUE HEADING 357.22, DIS-  
TANT TRUE HEADING 177.17 DEGREE  
STATUTE MILES 97.9, KILOMETERS  
157.55

You can use this example to check out your system, once loaded, and verify proper operation on your computer. One note: Distance is computed from the center of the grid square.

**“(The program)  
was a much welcome  
aid during microwave  
contest operations.”**

#### Versatile

The program will run on virtually any BASIC operation system. At present, it is used on a Tandy 100 laptop, a Kaypro CPM 2x, and some MS-DOS machines. For use on the Tandy 100 laptop, line 620 is deleted and a new line 620 is inserted as follows: (620 CLS).

This program allowed me to determine exact compass headings. It was a much welcome aid to help me aim my antenna towards the locations of unexpected stations that crop up during contest operation.

Normally I would have made available a disk or cassette, but due to the variety of computers being used—all taking different formats for input—I cannot reproduce a copy in your format. However, I will gladly send the program via modem. This seems to be the best method at present. WA5BNH has made the program fit into a Tandy PC-4 pocket computer with some re-assembling. Leon will make the PC-4 program available with or without the PC-4. Contact Leon Helms 10153, Ambassador Dr., San Diego CA 92126, for details. **73**

*Those microwave enthusiasts interested in obtaining more information about the myriad of 1 GHz and above operations going on in California, and around the world, may contact the author at the San Diego Microwave Group, 6345 Badger Lake, San Diego CA 92119. Chuck WB6IGP also now writes a monthly column for VHF and above activities “Above and Beyond.”*